

Kusto query to know the Shrink potential

Last updated by | Charlene Wang | Jan 3, 2023 at 12:30 AM PST

Contents

- [Issue summary -](#)
- [Mitigation steps -](#)
- [Classification](#)

Issue summary -

Space issue is a general reason for customer to update the database max size. This is a "How to" TSG which gives a kusto query to identify shrink potential of a database.

Mitigation steps -

```
let SqlInstanceName = 'b2456c1aa203';
let TenantRingName = 'tr3.northeurope1-a.worker.database.windows.net';
let lookBack = ago(7h);
let shrinkBuffer = 1.1;
MonDmloVirtualFileStats
| where originalEventTimestamp > lookBack
| project-rename tenant_ring_name = ClusterName, sql_instance_name = AppName, logical_server_name =
LogicalServerName, db_name = db_name
| where tenant_ring_name == TenantRingName
| where sql_instance_name == SqlInstanceName
| where is_primary_replica == 1
| where type_desc == 'ROWS' // logs _typically_ do not need to be shrunk
| project tenant_ring_name, file_id, sql_instance_name, NodeName, logical_server_name, db_name, type_desc,
originalEventTimestamp, logical_database_id, spaceused_gb = round(spaceused_mb / 1024.0, 2), max_size_mb,
size_on_disk_Gb = round(size_on_disk_bytes / 1024.0 / 1024 / 1024, 2)
| summarize arg_max(originalEventTimestamp, *) by NodeName, logical_server_name, db_name, type_desc
| extend
    free_percentage = round(iif(spaceused_gb == 0, 0.0, 100 - spaceused_gb / (size_on_disk_Gb / 100)), 2)
    , shrinkPotentialGb = size_on_disk_Gb - spaceused_gb
| join kind= leftouter (
    MonAnalyticsDBSnapshot
    | where TIMESTAMP > lookBack
    | where sql_instance_name == SqlInstanceName
    | where tenant_ring_name == TenantRingName
    | project-rename db_name = logical_database_name
    | summarize arg_max(TIMESTAMP, service_level_objective, physical_database_id) by tenant_ring_name,
sql_instance_name, logical_server_name, db_name, logical_database_id
) on tenant_ring_name, sql_instance_name, logical_server_name, db_name, logical_database_id
| project-away tenant_ring_name1, sql_instance_name1, logical_server_name1, db_name1, logical_database_id1
```

```
| extend isWorthShrinking = (spaceused_gb * shrinkBuffer + 20 < size_on_disk_Gb)
| extend shrinkCommand = strcat('Invoke-DBCCShrinkFile -SqlInstanceName ', sql_instance_name, ' -
PhysicalDatabaseId ', physical_database_id, ' -FileId ', file_id, ' -TargetSize ', toint(spaceused_gb * 1024 *
shrinkBuffer))
| order by shrinkPotentialGb desc
```

Classification

Root cause path -

CRUD/Service Issue/Database

How good have you found this content?



-